Maxima 演習　　　　　　　　　　　　　　　　2008.4 名古屋大学工学研究科　中里

I. 関数電卓として使う

1. 演算子

+	和
-	差
*	積
/	商
**	冪
^	冪（ ** と同じ)
!	階乗

2. 優先順位を指定する括弧はすべて（ ）を用いる。[ ], { } は別の意味を持っているので用いない。
3. 定数

%pi	円周率
%e	自然対数の底
%i	虚数
inf	∞
minf	−∞

4. 関数

abs(x)	絶対値
sqrt(x)	平方根
exp(x)	指数関数
log(x)	自然対数
sin(x), cos(x), tan(x)	三角関数　角度の単位はラジアン
asin(x), acos(x), atan(x)	逆三角関数
atan2(y,x)	点(x,y)の x 軸からの角度
sinh(x), cosh(x), tanh(x)	双曲線関数

5. 入力の最後に ; を付けると結果が表示される
   入力の最後に $ を付けると結果は表示されない
   xMaxima では入力の最後に ;　$ いずれも付けないと自動的に ; が付く
6. 数値評価するには float(式)

(4*5+2)/7	結果：22/7
float((4*5+2)/7)	結果：3.142857142857143

   ,numer としても良い

sqrt(2),numer	結果：1.4141213562373095

7. 有効数字の数 n を指定して数値計算するには
   fpprec:n
   bfloat(式)
   有効数字の数を確認するには
   fpprec
8. 結果を常に有効数字の数 n で表示する
   fpprintprec:n

演習問題

1. 次を試してみよう
   (1) 足算　　　　　　　　　　1+2;
   (2) 冪乗　　　　　　　　　　2^4;
   (3) 割算　　　　　　　　　　36/3;
   (4) 掛算　　　　　　　　　　5*9;
   (5) 優先順位を指定した計算　(6*x+9*x)/3;
   (6) %は1つ前に得られた結果を表す
   　　　　　　　　　　　　　　%-5;
   (7) !は階乗を表す　　　　　5!;
   (8) 平方根　　　　　　　　　sqrt(16);
   (9) 複素数　　　　　　　　　(3+5*%i)+(1+9*%i);
   (10) 絶対値　　　　　　　　abs(1+3*%i);
   (11) 三角関数　　　　　　　sin(%pi/3);
   (12) 分数　　　　　　　　　1/3+2/5;
2. 次の値を求めよ
   (1) 2の常用対数
   (2) 直角三角形の直角を挟む2辺の長さが3と4のとき、他の1辺の長さ
   (3) 1-2i　(i は虚数) の絶対値
   (4) exp(1-2i) の実部と虚部
   (5) 1-2i の複素平面での角度
3. 3 の平方根を有効数字20桁で求めよ
4. (-1+sqrt(2)*%i)/2 を3乗して1となることを確かめよ
5. 次の結果を比較し、違いの理由を考えよ。
   sqrt(5);
   sqrt(5.0);

II. 変数、値、評価

1. 変数　　　a, f_n, n123
   アルファベット、_ からなる文字
   最初の文字以外は数字を使うことができる
   大文字、小文字を区別
2. 変数は値をもつ
   変数の値を用いて計算が行われることを評価という
3. 変数に値を入れるには : を用いる
   a:3　　　　　　a の値 3
   このとき、:の右は評価される
   a:3
   fn:a^2+1　　　　fn の値 10
   値が定義されていないときには変数名が入る
   fn:x^2+1　　　　fn の値 x^2+1
   評価は入力された時点の 1 回のみ行われる
4. 変数の値を解除するには kill(変数)を用いる
   a:3
   fn:a^2+1　　　　fn の値 10
   kill(a)
   fn:a^2+1　　　　fn の値 a^2+1
   すべての変数の値を解除するには kill(all)
5. 値が定義されている変数を確認するには
   values;
6. 式を評価するには　ev(式) を用いる
   a:3
   b:4
   ev(a+b)　　　　　結果：7
   ev(式,変数=値)
   kill(all)
   ev(a+b,a=2)　　　結果：b+2
   ev(a+b,a=2,b=-4)　結果：-2
   ev(式,numer) 数値評価
7. 入力が ev(式)のみのときには ev()を省略できる
   a:3
   b:4
   a+b　　　　　　　結果：7
   a+b,a=2,b=-4　　　結果：-2
8. 名前の前に ' をつけるとその変数は評価されない
   a:3
   a　　　　　　　　結果：3
   'a　　　　　　　　結果：a
9. 変数の前に '' をつけるとその変数は評価される
   kill(a)
   fn:a^2+1　　　　fn の値 a^2+1
   a:3
   fn　　　　　　　結果：a^2+1
   ''fn　　　　　　　結果：10

演習問題

1. 次を試してみよう

(1)
```
a:5;
a*2;
```

(2)
```
aul:12;
aul*3;
values;
kill(aul);
aul;
values;
```

(3)
```
kill(all);
y:a*x^2;
a:b+1;
y;
''y;
a:b-1;
''y;
y:a*x^2;
a:b+1;
''y;
y:'a*x^2;
a:b-1;
''y;
```

III. 関数

1. 関数の定義
   f(x):=a*x^2+b*x+c
2. 関数の定義時には何も評価されない（コピーされるだけ）
   評価は関数が呼ばれたときに毎回行われる
3. 定義時に評価したいときには define を用いる
   define(f(x),a*x^2+b*x+c)
   評価は定義された時点の１回のみ行われる

演習問題

1.次を試してみよう

(1)	多項式の展開	expand((x^2+x+1)*(x^2-x+1));
(2)	多項式の係数	ratcoef(a*x^2+b*x+c+d*x^2,x^2);
(3)	多項式の因数分解	factor(2*x^2-5*x+2);
(4)	素因数分解	factor(2520);
(5)	有理数の通分	ratsimp(2/(x+1)-1/(x-1));
(6)	通分したときの分子	num(ratsimp(2/(x+1)-1/(x-1)))
(7)	通分したときの分母	denom(ratsimp(2/(x+1)-1/(x-1)))
(8)	有理式の部分分数展開	partfrac((x^3+2*x^2-4*x-3)/(x+1),x);

2.次を試してみよう

(1)
```
f(x):=x^2;
f(5);
f(x+y);
plot2d(f(x),[x,-2,2]);
```
(2)
```
kill(all);
y:x^2;
f1(x):=x+y;
define(f2(x),x+y);
f1(2,3);
f2(2,3);
```

3.(1) 因数 a, b, c, x, y, z を与えるとそれらの成分をもつ２つの３次元ベクトル (a,b,c),(x,y,z) がなす角度を返す関数 s(a,b,c,x,y,z)を定義せよ。

(2) A(0,0,0), B(1/2,1/2,1/2), C(1,0,1)としたときベクトル $\overrightarrow{AB}$ と $\overrightarrow{AC}$ のなす角度を求めよ。

IV. リスト

1. リストは要素の集まり
   例　[2,4,a,x^2-1]
2. リストＬのｎ番目の要素はL[n]
   L:[2,4,a,x^2-1]
   L[3]　 結果： a
   M:[[3,-1],[2,5]]
   M[1][2]　結果： -1
3. リストに関する関数
   length(L)　Ｌの要素数
   makelist(計算式、カウンタ変数、初期値、終値)
     makelist(i^2,i,1,5)　 結果： [1,4,9,16,25]
   makelist(計算式、変数、リスト)
   cons(要素,L)　　　　　Ｌの先頭に要素を加える
   endcons(要素,L)　　　 Ｌの最後に要素を加える
   append(L1,L2,...)　　　連結
   rest(L,n)　　　　　　 Ｌの最初のｎ個の要素を除いたリスト
   rest(L,-n)　　　　　　Ｌの最後のｎ個の要素を除いたリスト
   sort(L)　　　　　　　 Ｌを昇順に並べる
   reverse(L)　　　　　　反転
4. f([式1,式2,...]) は [f(式1),f(式2),...]
   sqrt([1,2,3])　　　　　結果： [1,sqrt(2),sqrt(3)]

演習問題

1.次を試してみよう
  (1) 5+[10,20,30,40];
  (2) 5*[10,20,30,40,50];
  (3) [10,20,30,40,50]^2;
  (4) append([10,20,30,40],[1,2,3,4]);
  (5) [10,20,30,40]+[1,2,3,4];
  (6) [10,20,30,40]*[1,2,3,4];
  (7) [10,20,30,40][3];
  (8) reverse([10,20,30,40,50]);
  (9) sort([60,50,30,40]);
  (10) makelist([n,n^2,n^3],n,1,5):
  (11) makelist(makelist(n^a,a,1,3),n,1,5);
  (12) makelist(x=y,y,[a,b,c]);
2.(1) x^2 の x が1から10までのリストを作成せよ
  (2) x^3 の x が1から10まで0.5置きのリストを作成せよ

V. 方程式を解く

1. 方程式を解くには solve(式,変数)
   結果はリストで与えられる
   ```
   solve(a*x^2+b*x+c=0,x)
   ```
   結果：[x=-(sqrt(b^2-4*a*c)+b)/(2*a), x=(sqrt(b^2-4*a*c)-b)/(2*a)]
   ```
   solve([2*x+3*y=1,-x+5*y=-2],[x,y])
   ```
   結果：[[x=11/13,y=-3/13]]
2. 解をその後の計算で使用するには
   ev()を用いる
   ```
   s:solve(a*x^2+b*x+c=0,x)
   ev(1/x,s[1])
   ```
   式の左辺を取り出す関数 lhs(e1=e2) 結果: e1
   式の右辺を取り出す関数 rhs(e1=e2) 結果: e2 を用いる
   ```
   1/rhs(s[1])
   ```
3. 次の違いを理解しよう
   ```
   f(x):=rhs(x)
   define(f(x),rhs(x))
   ```
4. 近似解 Newton 法　newton(零点を求める関数,検索開始値)
   ```
   load(newton)
   newton(exp(x)-3+x^2,0.8)
   ```

演習問題

1. x^3+3*x-7=0 の解を求めよ。また、有効数字 3 桁の数値で表せ
2. x^2+y^2=16, x^2-4=y の解を求めよ
3. ３点(0,0),(2,2),(3,1)を通る円の方程式 x^2+y^2+a*x+b*y+c=0 の係数 a,b,c を求めよ
   この円上の点で x=0.5 のときの y の値を求めよ
4. 方程式 x^6-6*x^4+13*x^2-2=0 の実数解を有効数字 5 桁まで求めよ
   ヒント：まず plot2d でグラフを描き、おおよその解を求め、newton 法でより正確な値を求める

VI. 微分と積分

1. 極限
   limit(関数,変数,近づける値)
   limit(関数,変数,近づける値,plus)　右極限
   limit(関数,変数,近づける値,minus) 左極限
2. １階微分 diff(式,変数)
   diff(sin(a*x),x)　結果：a*cos(a*x)
3. n 階微分 diff(式、変数、n)
   diff(sin(a*x),x,2)　結果：-a^2*sin(a*x)
   diff(sin(a*x),x,2,a,1)　結果：-2*a*sin(a*x)-a^2*x*cos(a*x)
4. 級数の和、積
   数値計算
   sum(関数,添え字変数,初期値,終値)
   sum(i^2,i,1,10)　結果：385
   product(関数,添え字変数,初期値,終値)
   product(i^2,i,1,5)　結果：14400
   記号的処理
   nusum(関数,添え字変数,初期値,終値)
   nusum(k^3,k,1,n)　結果：n^2*(n+1)^2/4
5. 不定積分 integrate(式,変数)
   integrate(1/(1+x),x)　結果： log(1+x)
6. 定積分 integrate(式,変数,下限,上限)
   integrate(1/(1+x),x,0,1)　結果： log(2)
7. 数値積分 simpson(関数名,初期値,終了値,等分数)
   load(simpsn)
   frenselsin(x):=sin(x^2/2)
   simpson(frenselsin,0,2,100)

演習問題

1. sin(x)/x の x が 0 での極限値を求めよ
2. x の増加する方向から π/2 に近づくとき、tan(x)の値を求めよ
3. x の減少する方向から π/2 に近づくとき、tan(x)の値を求めよ
4. x^x の微分を求めよ
5. x^2*sin(x) の微分を求めよ
6. sin(x)^2 の積分を求めよ
7. 1/(x^3+1)の積分を求めよ
8. x^2 の x が 0 から 1 までの積分値を求めよ
9. sin(x)/x の x が 0 から π までの積分値を求めよ
10. x^2*y^3 を x,y について積分せよ

VII. 微分方程式

1. 初期値を atvalue(関数,変数名=値,関数の値)
   微分方程式を解く desolve(微分方程式,求める関数)
```
atvalue(x(t),t=0,h)
atvalue(diff(x(t),t),t=0,0)
desolve(m*diff(x(t),t,2)=m*g-r*diff(x(t),t),x(t))
```
2. 2階までの微分方程式であれば ode2(微分方程式,求める関数,独立変数)
   次の結果を比較してみよう
```
desolve(diff(f(x),x)+x*f(x)=sin(x)/x,f(x))
ode2(diff(f(x),x)+x*f(x)=sin(x)/x,f(x),x)
```

演習問題

1. 次を試してみよう
   (1) desolve(diff(y(x),x)=x^2,y(x));
   (2) atvalue(y(x),x=0,1);
       desolve(diff(y(x),x)=x^2,y(x));
2. 次の微分方程式を解け
```
x'=2*x-y
y'=x+2y
x(0)=y(0)=1
```
   座標(x,y)の原点からの距離と x 軸とのなす角度を求めよ

VIII. プロット

1. 2次元プロット　plot2d(関数式,[定義変数名,開始値,終了値],オプション)
   範囲を100の区間に等分して計算し点をつなぐ
   plot2d(1/(x-1)+1,[x,-2,2])
   y軸の範囲を指定することもできる
   plot2d(1/(x-1)+1,[x,-2,2],[y,-5,5])
   複数のグラフを重ねてプロットすることもできる
   plot2d([cos(x),1-x^2/2],[x,-2*%pi,2*%pi])
2. 媒介変数を用いた2次元プロット
   plot2d([parametric,x成分の関数,y成分の関数,[媒介変数名,開始値,終了値],オプション)
   plot2d([parametric,cos(t),sin(t)],[t,0,2*%pi])
   媒介変数の範囲を10の区間に等分して計算、点をつなぐ
   区分数を変更するには [nticks,区分数]のオプションをつける
   plot2d([parametric,cos(t),sin(t)],[t,0,2*%pi],[nticks,100])
3. 点の組(x,y)が与えられている場合
   x:[10,20,30,40,50]
   y:[0.6,0.9,1.1,1.3,1.4]
   plot2d([discrete,x,y])
   点のみ打つ
   plot2d([discrete,x,y],[style,points])
4. 3次元プロット
   plot3d(関数式,[x成分変数名,開始値,終了値],[y成分変数名,開始値,終了値],オプション)
   plot3d(exp(-x^2-y^2),[x,-1,1],[y,-1,1])
   定義範囲を30x30の区間に等分して計算、四辺形で埋める
   区間数を変更するには[grid,x区間数,y区間数]のオプションをつける
   plot3d(exp(-x^2-y^2),[x,-1,1],[y,-1,1],[grid,50,50])
5. 媒介変数を用いた3次元プロット
   plot3d([x(s,t),y(s,t),z(s,t)],[s,開始値,終了値],[t,開始値,終了値],オプション)

演習問題

1. 次を試してみよう
   (1) sin(x)をxが0から2π までプロットする
   (2) sin(x)と2cos(x)をxが0から2π　まで同一グラフ上にプロットする
   (3) tをパラメータとしてx=sin(2t),y=cos(3t)をtが0から2π　までプロットする
   (4) sin(x+sin(y)) を xが0から4π まで、yが0から4π までプロットする
2. アステロイド x^(2/3)+y^(2/3)=1 をプロットせよ

IX. プログラム

1. block([局所変数リスト],手続き,return(結果))
2. 条件分岐
   if 判定 then 手続き else 手続き
   判定
     =, #, <, >, >=, <=
     整数? integerp()
     偶数? evenp()
     奇数? oddp()
     and, or
3. 繰り返し
   for カウンタ名 in リスト do(反復実行手続き)
     for i in [2,4,6] do(x:x+i^2)
   for カウンタ名:初期値 step 増分 thru 終了値 do(反復実行手続き)
   step 増分が省略されると増分=1 が用いられる
     for i:2 step 2 thru 6 do(x:x+i^2)
   for カウンタ名:初期値 step 増分 while 条件式 do(反復実行手続き)
     for i:2 step 2 while (i<=6) do(x:x+i^2)
   for カウンタ名:初期値 step 増分 unless 条件式 do(反復実行手続き)

演習問題

1. (1) 1^m+2^m+3^m+...+n^m を求める関数を作成せよ。
   (2) m=2, n=5 の場合について正しい答えが出ることを確かめよ
   (3) m=0, n=5 の場合について正しい答えが出ることを確かめよ
   (4) m=-2, n=5 の場合について正しい答えが出ることを確かめよ
2. (1) f(x)の局小値を求める関数を作成せよ
   (2) y=x^3-3*x+1 が極小となる x,y を求めよ
3. (1) ２分割法により有効数字３桁で f(x)=0 の解を求める関数を作成せよ
   答えはリストで返すものとする
   (2) x*sin(x)=1 の解を x が(0,10)の範囲ですべて求めよ

X. ベクトルと行列

1. ベクトルはリストで表す
```
v1:[1,0]
```
行列はmatrix()で定義
```
m1:matrix([2,-1],[2,1])
```
2. 加減・スカラー積が計算できる
```
v1:[1,0]
v2:[2,1]
-2*v1+v[2]/3     結果: [-4/3,1/3]
m1:matrix([1,-2],[2,1])
m2:matrix([2,-1],[1,2])
-2*m1+m2/3
```
3. ベクトルの内積は . を用いる
```
v1.v2
```
行列の積も . を用いる
```
m1.m2
```
行列の累乗は ^^ を用いる
```
m1^^10
```
4. 特殊な行列
```
ident(n)          n x n 単位行列
zeromatrix(m,n)   m x n 零行列
```
5. 行列に関する関数
```
transpose(M)     転置行列
adjoint(M)       余因子行列
determinant(M)   行列式
invert(M)        逆行列
mattrace(M)      行列のトレース
eigenvalues(M)   固有値
eigenvectors(M)  固有関数
```

演習問題

1. 次を試してみよう
```
(1) matrix([b,c],[d,e]).[x,y];
(2) matrix([b,c],[d,e]).matrix([x1,x2],[y1,y2]);
(3) [x,y].[x,y];
(4) m1=matrix([a,b],[c,d]);
    transpose(m1);
    determinant(m1);
    invert(m1);
```
2. 対称行列Mに対し、sin(M*t)を求める関数を作成せよ。
   ヒント：Mの固有関数・固有値を求めて計算する

XI. ファイルへの入出力

1. 計算手順を外部ファイルに記録
```
save("ファイル名",all);
```
特定の計算セルだけを保存
```
save("ファイル名",%in,%on,...);
```
読み込む
```
loadfile("ファイル名");
```
実行
```
playback(all);
```
部分的に実行
```
playback([開始セル番号,終了セル番号]);
```
2. データの入出力
```
read_list("ファイル名");
read_matrix("ファイル名");
write_data(リスト名,"ファイル名");
```
3. ファイルの中身の表示
```
prinftfile("ファイル名");
```
4. ファイル名はフルパスで指定する
```
"c:/maxima/data1"
```
/を用いることに注意(¥ではない)

演習問題

データ点数 n=701 の X 線回折のパターンのデータがファイル data にある。

(1) data を読み込み、リスト d に代入し、プロットせよ。

プロットしてみると，かなり雑音が含まれているので、畳み込みを利用してデータの平滑化を行う。畳み込みをするとき，もう 1 つの分布としてガウス分布を利用する。

$$f(i)=d(i)g_0+\sum_{j=1}^{i-1}d(i-j)g(j)+\sum_{j=1}^{n-i}d(i+j)g(j)$$

g(j)がガウス分布であれば，g_0，g(1)，g(2)の順に小さくなるので，f(i)は，i 番目のデ-タを中心に，その周辺のデ-タに重みをつけて足しあわせたものになることが分かる。したがって，雑音が低減されて，データが平滑化される。

(2) ガウス分布 g(j)として，データ点と同数のリストを作成せよ。ガウス関数は，規格化しておく必要がある。

$h_0=1$,　$h(j)=\exp\left(-200\left(\dfrac{j}{n}\right)^2\right)$,　j=1, . . . , n−1

$c=h_0+2\sum_{i=1}^{n-1}h(i)$,

$g_0=1/c$,　$g(j)=h(j)/c$,　j=1, . . . , n−1

(3) データと g の畳み込み f(i)を計算しプロットせよ。どのような分布をもつガウス関数を掛ければ良い結果が得られるか検討せよ。

XII. 高度な計算

1. フーリエ変換
   区間 [-p,p]　fourier(f,x,p);
   区間 [0,p]　fourcos(f,x,p); foursin(f,x,p);
   区間 [minf,inf] fourint(f,x);
   区間 [0,inf] fourintcos(f,x); fourintsin(f,x);
2. ラプラス変換
   laprace(expr, t, s);
   逆ラプラス変換
   ilt(expr,s,t);
3. 特殊関数
   load("orthopoly")
   ベッセル関数、Hermite 関数、ガンマ関数、erf 関数、Legendre 関数、
   spherical harmonic 関数、等

演習問題

1. (1) exp(2*t+a)*sin(t)*t をラプラス変換せよ
   (2) 上の結果を逆ラプラス変換して元に戻ることを確かめよ
2. 右図のオペアンプによるボルテージ・フォロワを考える。
   オペアンプの電圧利得を A とすると、V_{out} と V_{in} の関係はラプラス変換で次のように与えられる。

$$V_{out}(s)=\frac{A(s)}{1+A(s)}V_{in}(s)$$



   ここでオペアンプの電圧利得が次で与えらるとする。

$$A(s)=\frac{A_0}{\left(1+\frac{s}{\omega_1}\right)\left(1+\frac{s}{\omega_2}\right)}$$

   ここに $s = j\omega$（j は虚数、角振動数 $\omega = 2\pi f$、f は周波数）、A_0=1000、ω_1=1 とする。
   以下の問い に対し次の場合について答えよ。
   (a) $\omega_2 = 1$ (b) $\omega_2 = 10$ (c) $\omega_2 = 100$ (d) $\omega_2 = 1000$
   (1) A の絶対値と A の位相を角振動数 ω の関数としてプロットせよ。
   (2) A の絶対値が 1 のときの A の位相を ϕ としたとき、180° $-\phi$ をオペアンプの位相余裕という。
   位相余裕を計算せよ。
   (3) 角振動数 ω が0から まで変化したときの A の(実部,虚部)の軌跡をプロットせよ。
   (4) ボルテージ・フォロワの入力電圧 V_{in} が時間 t = 0 において 0V から 1V にステップ状に変化したとき出力電圧 V_{out} は時間とともにどのように変化するか。プロットせよ。